

# Frequency Domain Steerable Pyramid Filter Design

First derivative,  $k = 2$  filter

Ken Castleman 5/28/98

StPyr\_02.MCD

$$N := 16 \quad a := \text{floor}\left(\frac{N}{2}\right) \quad i := 0..N \quad k := 0..N \quad j := \sqrt{-1}$$

Define smooth-edged lowpass and highpass functions (raised cosine).

$$LP(x1, x2, x) := \text{if}\left[x < x1, 1, \text{if}\left[x > x2, 0, \sqrt{0.5 \cdot \left[1 + \cos\left[\pi \cdot \frac{x - x1}{x2 - x1}\right]\right]}\right]\right]$$

$$HP(x1, x2, x) := \text{if}\left[x < x1, 0, \text{if}\left[x > x2, 1, \sqrt{0.5 \cdot \left[1 - \cos\left[\pi \cdot \frac{x - x1}{x2 - x1}\right]\right]}\right]\right]$$

$$\theta_{i,k} := \text{angle}(i + 0.001 - a, k - a)$$

$$\rho_{i,k} := \sqrt{(i - a)^2 + (k - a)^2}$$

$$f1 := 0 \cdot a \quad f2 := \frac{5}{8} \cdot a \quad f3 := \frac{7}{8} \cdot a \quad f4 := 1.2 \cdot a$$

Define the transfer functions of the highpass, two lowpass and two bandpass filters.

The constants  $f1$  and  $f2$  control the steepness of the cutoffs;  $a$  is the folding frequency.

The  $LP(f3, f4, f)$  filter is not required for invertibility, but it makes the bandpass kernels smaller.

$$B1_{i,k} := LP(f3, f4, \rho_{i,k}) \cdot HP\left(f1, \frac{a}{2}, \rho_{i,k}\right) \cdot \cos(\theta_{i,k}) \quad B2_{i,k} := LP(f3, f4, \rho_{i,k}) \cdot HP\left(f1, \frac{a}{2}, \rho_{i,k}\right) \cdot \cos\left(\theta_{i,k} - \frac{\pi}{2}\right)$$

$$L1_{i,k} := LP\left(f1, \frac{a}{2}, \rho_{i,k}\right)$$

$$L0_{i,k} := LP(f2, a, \rho_{i,k})$$

$$H0_{i,k} := HP(f2, a, \rho_{i,k})$$

Compute functions to show that the constraints are satisfied.


$$M1_{i,k} := (B1_{i,k})^2 + (B2_{i,k})^2 \quad M2_{i,k} := (H0_{i,k})^2 + (L0_{i,k})^2$$

$$M3_{i,k} := \left[ (L1_{i,k})^2 + (B1_{i,k})^2 + (B2_{i,k})^2 \right]$$


$$M4_{i,k} := (H0_{i,k})^2 + (L0_{i,k})^2 \cdot \left[ (L1_{i,k})^2 + (B1_{i,k})^2 + (B2_{i,k})^2 \right]$$

$M4 = 1$  is Simoncelli's constraint.


Plot  $L_0(u, v)$ ,  $H_0(u, v)$  and the sum of their squared magnitudes



L0




H0




M2


Plot the two steerable bandpass filters and their sum of squared magnitudes.



B1




B2




M1


Plot  $L_1(u,v)$ , the sum of the bandpass squared magnitudes, their sum, and the overall transfer function.




L1



M1



M3




M4


**Demonstrate steerability** - generate and synthesize 60 degree filters and compare them.

$$B_{3_{i,k}} := LP(f3, f4, \rho_{i,k}) \cdot HP\left(f1, \frac{a}{2}, \rho_{i,k}\right) \cdot \cos\left(\theta_{i,k} - \frac{\pi}{3}\right)$$


$$B4 := \frac{1}{2} \cdot B1 + \frac{\sqrt{3}}{2} \cdot B2$$



B3



B4




B4 - B3

Compute the convolution kernels using the centered DFT.


$$W_{i,k} := \frac{1}{N+1} \cdot \exp\left[-j \cdot 2 \cdot \pi \cdot (i-a) \cdot \frac{k-a}{N+1}\right]$$

$$b1 := W \cdot (j \cdot B1) \cdot W \quad b2 := W \cdot (j \cdot B2) \cdot W \quad l0 := W \cdot L0 \cdot W \quad l1 := W \cdot L1 \cdot W \quad h0 := W \cdot H0 \cdot W$$

Plot the two bandpass filter impulse responses.




b1




b2


Plot the highpass and the two lowpass filter impulse responses.



h0



l0



l1

$$\sum_i \sum_k l0_{i,k} = 1 \quad \sum_i \sum_k l1_{i,k} = 1 \quad \sum_i \sum_k h0_{i,k} = 0 \quad \sum_i \sum_k b1_{i,k} = 0 \quad \sum_i \sum_k b2_{i,k} = 0$$

$$d := \begin{pmatrix} N+1 & N+1 \\ a & a \end{pmatrix} \quad d = \begin{pmatrix} 17 & 17 \\ 8 & 8 \end{pmatrix}$$

Round off to four digits and scale the values for writing kernel files.

b := 1

$$K2\_L0_{i,k} := \frac{\text{floor}(10000 \cdot \text{Re}(10_{i,k}) + 0.5)}{b}$$

$$K2\_L1_{i,k} := \frac{\text{floor}(10000 \cdot \text{Re}(11_{i,k}) + 0.5)}{b}$$

$$K2\_H0_{i,k} := \frac{\text{floor}(10000 \cdot \text{Re}(h0_{i,k}) + 0.5)}{b}$$

$$K2\_B1_{i,k} := \frac{\text{floor}(10000 \cdot \text{Re}(b1_{i,k}) + 0.5)}{b}$$

$$K2\_B2_{i,k} := \frac{\text{floor}(10000 \cdot \text{Re}(b2_{i,k}) + 0.5)}{b}$$

$$\sum_i \sum_k K2\_L0_{i,k} = 9996 \quad \sum_i \sum_k K2\_L1_{i,k} = 10005 \quad \sum_i \sum_k K2\_H0_{i,k} = 0 \quad \sum_i \sum_k K2\_B1_{i,k} = 0 \quad \sum_i \sum_k K2\_B2_{i,k} = 0$$

Write kernel files for input to the Wit 2-D convolution operator (Use b = 10,000 for unscaled kernels).

```
WRITEPRN(K2_L0_arr) := d      APPENDPRN(K2_L0_arr) := K2_L0
WRITEPRN(K2_L1_arr) := d      APPENDPRN(K2_L1_arr) := K2_L1
WRITEPRN(K2_H0_arr) := d      APPENDPRN(K2_H0_arr) := K2_H0
WRITEPRN(K2_B1_arr) := d      APPENDPRN(K2_B1_arr) := K2_B1
WRITEPRN(K2_B2_arr) := d      APPENDPRN(K2_B2_arr) := K2_B2
```

1. E. P. Simoncelli, and W. T. Freeman, "The Steerable Pyramid: A Flexible Architecture for Multi-Scale Derivative Computation," *Proc. ICIP-95*: 444-447, 1995.
2. P. J. Burt, and E. H. Adelson, "The Laplacian Pyramid as a Compact Image Code," *IEEE Trans. C-31*:532-540, 1983.
3. E. P. Simoncelli, W. T. Freeman, E. H. Adelson, and D. J. Heeger, "Shiftable Multiscale Transforms," *IEEE Trans. IT-38*(2):587-607, 1992.
4. W. T. Freeman and E. H. Adelson, "The Design and Use of Steerable Filters," *IEEE Trans PAMI-13*(9):891-906, 1991.

== END ==